
2 Jan van Eijck

1 Introduction7

What is a semantic theory, and why is it useful to implement semantic8

theories?9

In this chapter, a semantic theory is taken to be a collection of rules for10

specifying the interpretation of a class of natural language expressions. An11

example would be a theory of how to handle quantification, expressed as a set12

of rules for how to interpret determiner expressions like all, all except one, at13

least three but no more than ten.14

It will be demonstrated that implementing such a theory as a program that15

can be executed on a computer involves much less effort than is commonly16

thought, and has greater benefits than most linguists assume. Ideally, this17

Handbook should have example implementations in all chapters, to illustrate18

how the theories work, and to demonstrate that the accounts are fully explicit.19

What makes a semantic theory easy or hard to implement?20

What makes a semantic theory easy to implement is formal explicitness of21

the framework in which it is stated. Hard to implement are theories stated22

in vague frameworks, or stated in frameworks that elude explicit formulation23

because they change too often or too quickly. It helps if the semantic theory24

itself is stated in more or less formal terms.25

Choosing an implementation language: imperative versus declarative26

Well-designed implementation languages are a key to good software design,27

but while many well designed languages are available, not all kinds of language28

are equally suited for implementing semantic theories.29

Programming languages can be divided very roughly into imperative and30

declarative. Imperative programming consists in specifying a sequence of as-31

signment actions, and reading off computation results from registers. Declar-32

ative programming consists in defining functions or predicates and executing33

these definitions to obtain a result.34

Recall the old joke of the computer programmer who died in the shower?35

He was just following the instructions on the shampoo bottle: “Lather, rinse,36

repeat.” Following a sequence of instructions to the letter is the essence of37

imperative programming. The joke also has a version for functional program-38

mers. The definition on the shampoo bottle of the functional programmer39

runs:40

wash = lather : rinse : wash41

This is effectively a definition by co-recursion (like definition by recursion,42

but without a base case) of an infinite stream of lathering followed by rinsing43

followed by lathering followed by . . . .44

Implementing Semantic Theories 3

To be suitable for the representation of semantic theories, an implemen-45

tation language has to have good facilities for specifying abstract data types.46

The key feature in specifying abstract data types is to present a precise de-47

scription of that data type without referring to any concrete representation48

of the objects of that datatype and to specify operations on the data type49

without referring to any implementation details.50

This abstract point of view is provided by many-sorted algebras. Many51

sorted algebras are specifications of abstract datatypes. Most state-of-the art52

functional programming languages excel here. See below. An example of an53

abstract data type would be the specification of a grammar as a list of context54

free rewrite rules, say in Backus Naur form (BNF).55

Logic programming or functional programming: trade-offs56

First order predicate logic can be turned into a computation engine by adding
SLD resolution, unification and fixpoint computation. The result is called
datalog. SLD resolution is Linear resolution with a Selection function for
Definite sentences. Definite sentences, also called Horn clauses, are clauses
with exactly one positive literal. An example:

father(x) ∨ ¬parent(x) ∨ ¬male(x).

This can be viewed as a definition of the predicate father in terms of the
predicates parent and male, and it is usually written as a reverse implication,
and using a comma:

father(x)← parent(x),male(x).

To extend this into a full fledged programming paradigm, backtracking and cut57

(an operator for pruning search trees) were added (by Alain Colmerauer and58

Robert Kowalski, around 1972). The result is Prolog, short for programmation59

logique. Excellent sources of information on Prolog can be found at http:60

//www.learnprolognow.org/ and http://www.swi-prolog.org/.61

Pure lambda calculus was developed in the 1930s and 40s by the logician62

Alonzo Church, as a foundational project intended to put mathematics on63

a firm basis of ‘effective procedures’. In the system of pure lambda calculus,64

everything is a function. Functions can be applied to other functions to obtain65

values by a process of application, and new functions can be constructed from66

existing functions by a process of lambda abstraction.67

Unfortunately, the system of pure lambda calculus admits the formulation
of Russell’s paradox. Representing sets by their characteristic functions (essen-
tially procedures for separating the members of a set from the non-members),
we can define

r = λx · ¬(x x).

Now apply r to itself:68

4 Jan van Eijck

r r = (λx · ¬(x x))(λx · ¬(x x))

= ¬((λx · ¬(x x))(λx · ¬(x x)))

= ¬(r r).

So if (r r) is true then it is false and vice versa. This means that pure lambda69

calculus is not a suitable foundation for mathematics. However, as Church70

and Turing realized, it is a suitable foundation for computation. Elements of71

lambda calculus have found their way into a number of programming lan-72

guages such as Lisp, Scheme, ML, Caml, Ocaml, and Haskell.73

In the mid-1980s, there was no “standard” non-strict, purely-functional74

programming language. A language-design committee was set up in 1987, and75

the Haskell language is the result. Haskell is named after Haskell B. Curry, a76

logician who has the distinction of having two programming languages named77

after him, Haskell and Curry. For a famous defense of functional programming78

the reader is referred to Hughes (1989). A functional language has non-strict79

evaluation or lazy evaluation if evaluation of expressions stops ‘as soon as80

possible’. In particular, only arguments that are necessary for the outcome81

are computed, and only as far as necessary. This makes it possible to handle82

infinite data structures such as infinite lists. We will use this below to represent83

the infinite domain of natural numbers.84

A declarative programming language is better than an imperative pro-85

gramming language for implementing a description of a set of semantic rules.86

The two main declarative programming styles that are considered suitable for87

implementating computational semantics are logic programming and func-88

tional programming. Indeed, computational paradigms that emerged in com-89

puter science, such as unification and proof search, found their way into seman-90

tic theory, as basic feature value computation mechanisms and as resolution91

algorithms for pronoun reference resolution.92

If unification and first order inference play an important role in a semantic93

theory, then a logic programming language like Prolog may seem a natural94

choice as an implementation language. However, while unification and proof95

search for definite clauses constitute the core of logic programming (there is96

hardly more to Prolog than these two ingredients), functional programming97

encompasses the whole world of abstract datatype definition and polymorphic98

typing. As we will demonstrate below, the key ingredients of logic program-99

ming are easily expressed in Haskell, while Prolog is not very suitable for100

expressing data abstraction. Therefore, in this chapter we will use Haskell101

rather than Prolog as our implementation language. For a textbook on com-102

putational semantics that uses Prolog, we refer to Blackburn & Bos (2005). A103

recent computational semantics textbook that uses Haskell is Eijck & Unger104

(2010).105

Modern functional programming languages such as Haskell are in fact im-106

plementations of typed lambda calculus with a flexible type system. Such107

languages have polymorphic types, which means that functions and opera-108

Implementing Semantic Theories 5

tions can apply generically to data. E.g., the operation that joins two lists has109

as its only requirement that the lists are of the same type a — where a can110

be the type of integers, the type of characters, the type of lists of characters,111

or any other type — and it yields a result that is again a list of type a.112

This chapter will demonstrate, among other things, that implementing a113

Montague style fragment in a functional programming language with flexible114

types is a breeze: Montague’s underlying representation language is typed115

lambda calculus, be it without type flexibility, so Montague’s specifications116

of natural language fragments in PTQ Montague (1973) and UG Montague117

(1974b) are in fact already specifications of functional programs. Well, almost.118

Unification versus function composition in logical form construction119

If your toolkit has just a hammer in it, then everything looks like a nail. If120

your implementation language has built-in unification, it is tempting to use121

unification for the composition of expressions that represent meaning. The122

Core Language Engine Alshawi (1992); Alshawi & Eijck (1989) uses unification123

to construct logical forms.124

For instance, instead of combining noun phrase interpretations with verb125

phrase interpretations by means of functional composition, in a Prolog im-126

plementation a verb phrase interpretation typically has a Prolog variable X127

occupying a subjVal slot, and the noun phrase interpretation typically unifies128

with the X. But this approach will not work if the verb phrase contains more129

than one occurrence of X. Take the translation of No one was allowed to pack130

and leave. This does not mean the same as No one was allowed to pack and131

no one was allowed to leave. But the confusion of the two is hard to avoid132

under a feature unification approach.133

Theoretically, function abstraction and application in a universe of higher134

order types are a much more natural choice for logical form construction.135

Using an implementation language that is based on type theory and function136

abstraction makes it particularly easy to implement the elements of semantic137

processing of natural language, as we will demonstrate below.138

Literate Programming139

This Chapter is written in so-called literate programming style. Literate pro-140

gramming, as advocated by Donald Knuth in Knuth (1992), is a way of writing141

computer programs where the first and foremost aim of the presentation of a142

program is to make it easily accessible to humans. Program and documenta-143

tion are in a single file. In fact, the program source text is extracted from the144

LATEX source text of the chapter. Pieces of program source text are displayed145

as in the following Haskell module declaration for this Chapter:146

6 Jan van Eijck

module IST where

import Data.List

import Data.Char

import System.IO

147

This declares a module called IST, for “Implementing a Semantic Theory”,148

and imports the Haskell library with list processing routines called Data.List,149

the library with character processing functions Data.Char, and the input-150

output routines library System.IO.151

We will explain most programming constructs that we use, while avoiding152

a full blown tutorial. For tutorials and further background on programming153

in Haskell we refer the reader to www.haskell.org, and to the textbook Eijck154

& Unger (2010).155

You are strongly encouraged to install the Haskell Platform on your com-156

puter, download the software that goes with this chapter from internet address157

https://github.com/janvaneijck/ist, and try out the code for yourself.158

The advantage of developing fragments with the help of a computer is that159

interacting with the code gives us feedback on the clarity and quality of our160

formal notions.161

The role of models in computational semantics162

If one looks at computational semantics as an enterprise of constructing logical163

forms for natural language sentences to express their meanings, then this may164

seem a rather trivial exercise, or as Stephen Pulman once phrased it, an165

“exercise in typesetting”. “John loves Mary” gets translated into L(j,m),166

and so what? The point is that L(j,m) is a predication that can be checked167

for truth in an appropriate formal model. Such acts of model checking are168

what computational semantics is all about. If one implements computational169

semantics, one implements appropriate models for semantic interpretation as170

well, plus the procedures for model checking that make the computational171

engine tick. We will illustrate this with the examples in this Chapter.172

Implementing Semantic Theories 7

2 Direct Interpretation or Logical Form?173

In Montague style semantics, there are two flavours: use of a logical form174

language, as in PTQ Montague (1973) and UG Montague (1974b), and direct175

semantic interpretation, as in EAAFL Montague (1974a).176

To illustrate the distinction, consider the following BNF grammar for gen-
eralized quantifiers:

Det ::= Every | All | Some | No | Most.

The data type definition in the implementation follows this to the letter:177

data Det = Every | All | Some | No | Most

deriving Show
178

Let D be some finite domain. Then the interpretation of a determiner on179

this domain can be viewed as a function of type PD → PD → {0, 1}. In180

Montague style, elements of D have type e and the type of truth values is181

denoted t, so this becomes: (e → t) → (e → t) → t. Given two subsets p, q182

of D, the determiner relation does or does not hold for these subsets. E.g.,183

the quantifier relation All holds between two sets p and q iff p ⊆ q. Similarly184

the quantifier relation Most holds between two finite sets p and q iff p∩ q has185

more elements than p− q. Let’s implement this.186

Direct interpretation187

A direct interpretation instruction for “All” for a domain of integers (so now188

the role of e is played by Int) is given by:189

intDET :: [Int] -> Det

-> (Int -> Bool) -> (Int -> Bool) -> Bool

intDET domain All = \ p q ->

filter (\x -> p x && not (q x)) domain == []

190

Here, [] is the empty list. The type specification says that intDET is a191

function that takes a list of integers, next a determiner Det, next an integer192

property, next another integer property, and yields a boolean (True or False).193

The function definition for All says that All is interpreted as the relation194

between properties p and q on a domain that evaluates to True iff the set of195

objects in the domain that satisfy p but not q is empty.196

Let’s play with this. In Haskell the property of being greater than some197

number n is expressed as (> n). A list of integers can specified as [n..m]. So198

here goes:199

*IST> intDET [1..100] All (> 2) (> 3)200

False201

*IST> intDET [1..100] All (> 3) (> 2)202

True203

All numbers in the range 1..100 that are greater that 2 are also greater204

than 3 evaluates to False, all numbers s in the range 1..100 that are greater205

that 3 are also greater than 2 evaluates to True. We can also evaluate on206

infinite domains. In Haskell, if n is an integer, then [n..] gives the infinite207

list of integer numbers starting with n, in increasing order. This gives:208

IST> intDET [1..] All (> 2) (> 3)209

False210

*IST> intDET [1..] All (> 3) (> 2)211

...212

The second call does not terminate, for the model checking procedure is213

dumb: it does not ‘know’ that the domain is enumerated in increasing order.214

By the way, you are trying out these example calls for yourself, aren’t you?215

A direct interpretation instruction for “Most” is given by:216

intDET domain Most = \ p q ->

let

xs = filter (\x -> p x && not (q x)) domain

ys = filter (\x -> p x && q x) domain

in length ys > length xs

217

This says that Most is interpreted as the relation between properties p and218

q that evaluates to True iff the set of objects in the domain that satisfy both219

p and q is larger than the set of objects in the domain that satisfy p but not220

q. Note that this implementation will only work for finite domains.221

Translation into logical form222

To contrast this with translation into logical form, we define a datatype for223

formulas with generalized quantifiers.224

Building blocks that we need for that are names and identifiers (type Id),225

which are pairs consisting of a name (a string of characters) and an integer226

index.227

type Name = String

data Id = Id Name Int deriving (Eq,Ord)
228

What this says is that we will use Name is a synonym for String, and229

that an object of type Id will consist of the identifier Id followed by a Name230

followed by an Int. In Haskell, Int is the type for fixed-length integers. Here231

are some examples of identifiers:232

Implementing Semantic Theories 9

ix = Id "x" 0

iy = Id "y" 0

iz = Id "z" 0

233

From now on we can use ix for Id "x" 0, and so on. Next, we define terms.
Terms are either variables or functions with names and term arguments. First
in BNF notation:

t ::= vi | fi(t, . . . , t).

The indices on variables vi and function symbols fi can be viewed as names.234

Here is the corresponding data type:235

data Term = Var Id | Struct Name [Term] deriving (Eq,Ord)236

Some examples of variable terms:237

x = Var ix

y = Var iy

z = Var iz

238

An example of a constant term (a function without arguments):239

zero :: Term

zero = Struct "zero" []
240

Some examples of function symbols:241

s = Struct "s"

t = Struct "t"

u = Struct "u"

242

Function symbols can be combined with constants to define so-called243

ground terms (terms without occurrences of variables). In the following, we244

use s[ ] for the successor function.245

one = s[zero]

two = s[one]

three = s[two]

four = s[three]

five = s[four]

246

The function isVar checks whether a term is a variable; it uses the type247

Bool for Boolean (true or false). The type specification Term -> Bool says248

that isVar is a classifier of terms. It classifies the the terms that start with249

Var as variables, and all other terms as non-variables.250

isVar :: Term -> Bool

isVar (Var _) = True

isVar _ = False

251

The function isGround checks whether a term is a ground term (a term252

without occurrences of variables); it uses the Haskell primitives and and map,253

which you should look up in a Haskell tutorial if you are not familiar with254

them.255

isGround :: Term -> Bool

isGround (Var _) = False

isGround (Struct _ ts) = and (map isGround ts)

256

This gives (you should check this for yourself):257

*IST> isGround zero258

True259

*IST> isGround five260

True261

*IST> isGround (s[x])262

False263

The functions varsInTerm and varsInTerms give the variables that occur in264

a term or a term list. Variable lists should not contain duplicates; the function265

nub cleans up the variable lists. If you are not familiar with nub, concat and266

function composition by means of ·, you should look up these functions in a267

Haskell tutorial.268

varsInTerm :: Term -> [Id]

varsInTerm (Var i) = [i]

varsInTerm (Struct _ ts) = varsInTerms ts

varsInTerms :: [Term] -> [Id]

varsInTerms = nub . concat . map varsInTerm

269

We are now ready to define formulas from atoms that contain lists of terms.
First in BNF:

φ ::= A(t, . . . , t) | t = t | ¬φ | φ ∧ φ | φ ∨ φ | Qvφφ.

Here A(t, . . . , t) is an atom with a list of term arguments. In the implemen-270

tation, the data-type for formulas can look like this:271

data Formula = Atom Name [Term]

| Eq Term Term

| Not Formula

| Cnj [Formula]

| Dsj [Formula]

| Q Det Id Formula Formula

deriving Show

272

Equality statements Eq Term Term express identities t1 = t2. The Formula273

data type defines conjunction and disjunction as lists, with the intended mean-274

ing that Cnj fs is true iff all formulas in fs are true, and that Dsj fs is true275

iff at least one formula in fs is true. This will be taken care of by the truth276

definition below.277

Before we can use the data type of formulas, we have to address a syntactic278

issue. The determiner expression is translated into a logical form construction279

recipe, and this recipe has to make sure that variables bound by a newly280

introduced generalized quantifier are bound properly. The definition of the281

fresh function that takes care of this can be found in the appendix. It is used282

in the translation into logical form for the quantifiers:283

lfDET :: Det ->

(Term -> Formula) -> (Term -> Formula) -> Formula

lfDET All p q = Q All i (p (Var i)) (q (Var i)) where

i = Id "x" (fresh [p zero, q zero])

lfDET Most p q = Q Most i (p (Var i)) (q (Var i)) where

i = Id "x" (fresh [p zero, q zero])

lfDET Some p q = Q Some i (p (Var i)) (q (Var i)) where

i = Id "x" (fresh [p zero, q zero])

lfDET No p q = Q No i (p (Var i)) (q (Var i)) where

i = Id "x" (fresh [p zero, q zero])

284

Note that the use of a fresh index is essential. If an index i is not fresh,285

this means that it is used by a quantifier somewhere inside p or q, which286

gives a risk that if these expressions of type Term -> Formula are applied to287

Var i, occurrences of this variable may get bound by the wrong quantifier288

expression.289

Of course, the task of providing formulas of the form All v φ1φ2 or the290

form Most v φ1φ2 with the correct interpretation is now shifted to the truth291

definition for the logical form language. We will turn to this in the next292

Section.293

3 Model Checking Logical Forms294

The example formula language from Section 2 is first order logic with equality295

and the generalized quantifier Most. This is a genuine extension of first order296

logic with equality, for it is proved in Barwise & Cooper (1981) that Most is297

not expressible in first order logic.298

Once we have a logical form language like this, we can dispense with299

extending this to a higher order typed version, and instead use the implemen-300

tation language to construct the higher order types.301

Think of it like this. For any type a, the implementation language gives302

us properties (expressions of type a → Bool), relations (expressions of type303

a → a → Bool), higher order relations (expressions of type (a → Bool) →304

(a→ Bool)→ Bool), and so on. Now replace the type of Booleans with that305

of logical forms or formulas (call it F ), and the type a with that of terms (call306

it T ). Then the type T → F expresses an LF property, the type T → T → F307

an LF relation, the type (T → F ) → (T → F ) → F a higher order relation,308

suitable for translating generalized quantifiers, and so on.309

For example, the LF translation of the generalized quantifier Most in Sec-310

tion 2, produces an expression of type (T → F )→ (T → F )→ F .311

Tarski’s famous truth definition for first order logic (Tarski, 1956) has as312

key ingredients variable assignments, interpretations for predicate symbols,313

and interpretations for function symbols, and proceeds by recursion on the314

structure of formulas.315

A domain of discourse D together with an interpretation function I that316

interprets predicate symbols as properties or relations on D, and function317

symbols as functions on D, is called a first order model.318

In our implementation, we have to distinguish between the interpretation319

for the predicate letters and that for the function symbols, for they have320

different types:321

type Interp a = Name -> [a] -> Bool

type FInterp a = Name -> [a] -> a
322

These are polymorphic declarations: the type a can be anything. Suppose323

our domain of entities consists of integers. Let us say we want to interpret on324

the domain of the natural numbers. Then the domain of discourse is infinite.325

Since our implementation language has non-strict evaluation, we can handle326

infinite lists. The domain of discourse is given by:327

naturals :: [Integer]

naturals = [0..]
328

Implementing Semantic Theories 13

The type Integer is for integers of arbitrary size. Other domain definitions329

are also possible. Here is an example of a finite number domain, using the fixed330

size data type Int:331

numbers :: [Int]

numbers = [minBound..maxBound]
332

Let V be the set of variables of the language. A function g : V → D is333

called a variable assignment or valuation.334

Before we can turn to evaluation of formulas, we have to construct valua-335

tion functions of type Term -> a, given appropriate interpretations for func-336

tion symbols, and given an assignment to the variables that occur in terms.337

A variable assignment, in the implementation, is a function of type338

Id -> a, where a is the type of the domain of interpretation. The term lookup339

function takes a function symbol interpretatiomn (type FInterp a) and vari-340

able assigment (type Id -> a) as inputs, and constructs a term assignment341

(type Term -> a), as follows.342

tVal :: FInterp a -> (Id -> a) -> Term -> a

tVal fint g (Var v) = g v

tVal fint g (Struct str ts) =

fint str (map (tVal fint g) ts)

343

tVal computes a value (an entity in the domain of discourse) for any term,344

on the basis of an interpretation for the function symbols and an assigment345

of entities to the variables. Understanding how this works is one of the keys346

to understanding the truth definition for first order predicate logic, as it is347

explained in textbooks of logic. Here is that explanation once more:348

• If the term is a variable, tVal borrows its value from the assignment g for349

variables.350

• If the term is a function symbol followed by a list of terms, then tVal is351

applied recursively to the term list, which gives a list of entities, and next352

the interpretation for the function symbol is used to map this list to an353

entity.354

Example use: fint1 gives an interpretation to the function symbol s while355

(\ _ -> 0) is the anonymous function that maps any variable to 0. The result356

of applying this to the term five (see the definition above) gives the expected357

value:358

*IST> tVal fint1 (\ _ -> 0) five359

5360

The truth definition of Tarski assumes a relation interpretation, a function361

interpretation and a variable assigment, and defines truth for logical form362

expression by recursion on the structure of the expression.363

Given a structure with interpretation function M = (D, I), we can define364

a valuation for the predicate logical formulas, provided we know how to deal365

with the values of individual variables.366

Let g be a variable assignment or valuation. We use g[v := d] for the367

valuation that is like g except for the fact that v gets value d (where g might368

have assigned a different value). For example, let D = {1, 2, 3} be the domain369

of discourse, and let V = {v1, v2, v3}. Let g be given by g(v1) = 1, g(v2) =370

2, g(v3) = 3. Then g[v1 := 2] is the valuation that is like g except for the fact371

that v1 gets the value 2, i.e. the valuation that assigns 2 to v1, 2 to v2, and 3372

to v3.373

Here is the implementation of g[v := d]:374

change :: (Id -> a) -> Id -> a -> Id -> a

change g v d = \ x -> if x == v then d else g x
375

Let M = (D, I) be a model for language L, i.e., D is the domain of376

discourse, I is an interpretation function for predicate letters and function377

symbols. Let g be a variable assignment for L in M . Let F be a formula of378

our logical form language.379

Now we are ready to define the notion M |=g F , for F is true in M
under assignment g, or: g satisfies F in model M . We assume P is a one-place
predicate letter, R is a two-place predicate letter, S is a three-place predicate
letter. Also, we use [[t]]Ig as the term interpretation of t under I and g. With
this notation, Tarski’s truth definition can be stated as follows:

M |=g Pt iff [[t]]Ig ∈ I(P )
M |=g R(t1, t2) iff ([[t1]]Ig, [[t2]]Ig) ∈ I(R)
M |=g S(t1, t2, t3) iff ([[t1]]Ig, [[t2]]Ig, [[t3]]Ig) ∈ I(S)
M |=g (t1 = t2) iff [[t1]]Ig = [[t2]]Ig
M |=g ¬F iff it is not the case that M |=g F.
M |=g (F1 ∧ F2) iff M |=g F1 and M |=g F2

M |=g (F1 ∨ F2) iff M |=g F1 or M |=g F2

M |=g QvF1F2 iff {d |M |=g[v:=d] F1} and {d |M |=g[v:=d] F2}
are in the relation specified by Q

What we have presented just now is a recursive definition of truth for our380

logical form language. The ‘relation specified by Q’ in the last clause refers to381

the generalized quantifier interpretations for all, some, no and most. Here is382

an implementation of quantifiers are relations:383

Implementing Semantic Theories 15

qRel :: Eq a => Det -> [a] -> [a] -> Bool

qRel All xs ys = all (\x -> elem x ys) xs

qRel Some xs ys = any (\x -> elem x ys) xs

qRel No xs ys = not (qRel Some xs ys)

qRel Most xs ys =

length (intersect xs ys) > length (xs \\ ys)

384

If we evaluate closed formulas — formulas without free variables — the385

assignment g is irrelevant, in the sense that any g gives the same result. So386

for closed formulas F we can simply define M |= F as: M |=g F for some387

variable assignment g. But note that the variable assignment is still crucial388

for the truth definition, for the property of being closed is not inherited by389

the components of a closed formula.390

Let us look at how to implement an evaluation function. It takes as its391

first argument a domain, as its second argument a predicate interpretation392

function, as its third argument a function interpretation function, as its fourth393

argument a variable assignment, as its fifth argument a formula, and it yields394

a truth value. It is defined by recursion on the structure of the formula. The395

type of the evaluation function eval reflects the above assumptions.396

eval :: Eq a =>

[a] ->

Interp a ->

FInterp a ->

(Id -> a) ->

Formula -> Bool

397

The evaluation function is defined for all types a that belong to the class Eq.398

The assumption that the type a of the domain of evaluation is in Eq is needed399

in the evaluation clause for equalities. The evaluation function takes a universe400

(represented as a list, [a]) as its first argument, an interpretation function401

for relation symbols (Interp a) as its second argument, an interpretation402

function for function symbols as its third argument, a variable assignment403

(Id -> a) as its fourth argument, and a formula as its fifth argument. The404

definition is by structural recursion on the formula:405

eval domain i fint = eval’ where

eval’ g (Atom str ts) = i str (map (tVal fint g) ts)

eval’ g (Eq t1 t2) = tVal fint g t1 == tVal fint g t2

eval’ g (Not f) = not (eval’ g f)

eval’ g (Cnj fs) = and (map (eval’ g) fs)

eval’ g (Dsj fs) = or (map (eval’ g) fs)

eval’ g (Q det v f1 f2) = let

restr = [ d | d <- domain, eval’ (change g v d) f1 ]

body = [ d | d <- domain, eval’ (change g v d) f2 ]

in qRel det restr body

406

This evaluation function can be used to check the truth of formulas in407

appropriate domains. The domain does not have to be finite. Suppose we408

want to check the truth of “There are even natural numbers”. Here is the409

formula:410

form0 = Q Some ix (Atom "Number" [x]) (Atom "Even" [x])411

We need an interpretation for the predicates “Number” and “Even”. We412

also throw in an interpretation for “Less than”:413

int0 :: Interp Integer

int0 "Number" = \[x] -> True

int0 "Even" = \[x] -> even x

int0 "Less_than" = \[x,y] -> x < y

414

Note that relates language (strings like “Number”, “Even”) to predicates415

on a model (implemented as Haskell functions). So the function int0 is part416

of the bridge between language and the world (or: between language and the417

model under consideration).418

For this example, we don’t need to interpret function symbols, so any419

function interpretation will do. But for other examples we want to give names420

to certain numbers, using the constants “zero”, “s”, “plus”, “times”. Here is421

a suitable term interpretation function for that:422

fint0 :: FInterp Integer

fint0 "zero" [] = 0

fint0 "s" [i] = succ i

fint0 "plus" [i,j] = i + j

fint0 "times" [i,j] = i * j

423

Again we see a distinction between syntax (expressions like “plus” and424

“times”) and semantics (Haskell operations like + and *).425

Implementing Semantic Theories 17

*IST> eval naturals int0 fint0 (\ _ -> 0) form0426

True427

This example uses a variable assigment \ _ -> 0 that maps any variable428

to 0.429

Now suppose we want to evaluate the following formula:430

form1 = Q All ix (Atom "Number" [x])

(Q Some iy (Atom "Number" [y])

(Atom "Less_than" [x,y]))

431

This says that for every number there is a larger number, which as we all432

know is true on the natural numbers. But this fact cannot be established by433

model checking. The following computation does not halt:434

*IST> eval naturals int0 fint0 (\ _ -> 0) form1435

...436

This illustrates that model checking on the natural numbers is undecidable.437

Still, many useful facts can be checked, and new relations can be defined in438

terms of a few primitive ones.439

Suppose we want to define the relation “divides”. A natural number x440

divides a natural number y if there is a number z with the property that441

x ∗ z = y. This is easily defined, as follows:442

divides :: Term -> Term -> Formula

divides m n = Q Some iz (Atom "Number" [z])

(Eq n (Struct "times" [m,z]))

443

This gives:444

*IST> eval naturals int0 fint0 (\ _ -> 0) (divides two four)445

True446

The process of defining truth for expressions of natural language is sim-447

ilar to that of evaluating formulas in mathematical models. The differences448

are that the models may have more internal structure than mathematical449

domains, and that substantial vocabularies need to be interpreted.450

Interpretation of Natural Language Fragments451

Where in mathematics it is enough to specify the meanings of ‘less than’,452

‘plus’ and ‘times’, and next define notions like ‘even’, ‘odd’, ‘divides’, ‘prime’,453

‘composite’, in terms of these primitives, in natural language understanding454

there is no such privileged core lexicon. This means we need interpretations455

for all non-logical items in the lexicon of a fragment.456

Page: 17 job: VanEijck macro: handbook.cls date/time: 8-Apr-2014/23:10



18 Jan van Eijck

To give an example, assume that the domain of discourse is a finite set of457

entities. Let the following data type be given.458

data Entity = A | B | C | D | E | F | G

| H | I | J | K | L | M

deriving (Eq,Show,Bounded,Enum)

459

Now we can define entities as follows:460

entities :: [Entity]

entities = [minBound..maxBound]
461

Now, proper names will simply be interpreted as entities.462

alice, bob, carol :: Entity

alice = A

bob = B

carol = C

463

Common nouns such as girl and boy as well as intransitive verbs like laugh464

and weep are interpreted as properties of entities. Transitive verbs like love465

and hate are interpreted as relations between entities.466

Let’s define a type for predications:467

type Pred a = [a] -> Bool468

Some example properties:469

girl, boy :: Pred Entity

girl = \ [x] -> elem x [A,C,D,G]

boy = \ [x] -> elem x [B,E,F]

470

Some example binary relations:471

love, hate :: Pred Entity

love = \ [x,y] -> elem (x,y) [(A,A),(A,B),(B,A),(C,B)]

hate = \ [x,y] -> elem (x,y) [(B,C),(C,D)]

472

And here is an example of a ternary relation:473

Implementing Semantic Theories 19

give, introduce :: Pred Entity

give = \ [x,y,z] -> elem (x,y,z) [(A,H,B),(A,M,E)]

introduce = \ [x,y,z] -> elem (x,y,z) [(A,A,B),(A,B,C)]

474

The intention is that the first element in the list specifies the giver, the475

second element the receiver, and the third element what is given.476

Operations on predications477

Once we have this we can specify operations on predications. A simple example478

is passivization, which is a process of argument reduction: the agent of an479

action is dropped. Here is a possible implementation:480

passivize :: [a] -> Pred a -> Pred a

passivize domain r = \ xs -> any (\ y -> r (y:xs)) domain
481

Let’s check this out:482

*IST> :t (passivize entities love)483

(passivize entities love) :: Pred Entity484

*IST> filter (\ x -> passivize entities love [x]) entities485

[A,B]486

Note that this also works for for ternary predicates. Here is the illustration:487

*IST> :t (passivize entities give)488

(passivize’ entities give) :: Pred Entity489

*IST> filter (passivize entities give)490

[[x,y] | x <- entities, y <- entities]491

[[H,B],[M,E]]492

Reflexivization493

Another example of argument reduction in natural languages is reflexivization.494

The view that reflexive pronouns are relation reducers is folklore among logi-495

cians, but can also be found in linguistics textbooks, such as Daniel Büring’s496

book on Binding Theory (Büring, 2005, pp. 43–45).497

Under this view, reflexive pronouns like himself and herself differ seman-498

tically from non-reflexive pronouns like him and her in that they are not499

interpreted as individual variables. Instead, they denote argument reducing500

functions. Consider, for example, the following sentence:501

Alice loved herself. (1)

The reflexive herself is interpreted as a function that takes the two-place502

predicate loved as an argument and turns it into a one-place predicate, which503

takes the subject as an argument, and expresses that this entity loves itself.504

This can be achieved by the following function self.505

self :: Pred a -> Pred a

self r = \ (x:xs) -> r (x:x:xs)
506

Here is an example application:507

*IST> :t (self love)508

(self love) :: Pred Entity509

*IST> :t \ x -> self love [x]510

\ x -> self love [x] :: Entity -> Bool511

*IST> filter (\ x -> self love [x]) entities512

[A]513

This approach to reflexives has two desirable consequences. The first one514

is that the locality of reflexives immediately falls out. Since self is applied to515

a predicate and unifies arguments of this predicate, it is not possible that an516

argument is unified with a non-clause mate. So in a sentence like (2), herself517

can only refer to Alice but not to Carol.518

Carol believed that Alice loved herself. (2)

The second one is that it also immediately follows that reflexives in subject519

position are out.520

∗Herself loved Alice. (3)

Given a compositional interpretation, we first apply the predicate loved to521

Alice, which gives us the one-place predicate λ[x] 7→ love [x, a]. Then trying522

to apply the function self to this will fail, because it expects at least two523

arguments, and there is only one argument position left.524

Reflexive pronouns can also be used to reduce ditransitive verbs to transi-525

tive verbs, in two possible ways: the reflexive can be the direct object or the526

indirect object:527

Alice introduced herself to Bob. (4)

Bob gave the book to himself. (5)

The first of these is already taken care of by the reduction operation above.528

For the second one, here is an appropriate reduction function:529

self’ :: Pred a -> Pred a

self’ r = \ (x:y:xs) -> r (x:y:x:xs)
530

Quantifier scoping531

Quantifier scope ambiguities can be dealt with in several ways. From the532

point of view of type theory it is attractive to view sequences of quantifiers as533

functions from relations to truth values. E.g., the sequence “every man, some534

woman” takes a binary relation λxy·R[x, y] as input and yields True if and only535

if it is the case that for every man x there is some woman y for which R[x, y]536

holds. To get the reversed scope reading, just swap the quantifier sequence,537

and transform the relation by swapping the first two argument places, as538

follows:539

swap12 :: Pred a -> Pred a

swap12 r = \ (x:y:xs) -> r (y:x:xs)
540

So scope inversion can be viewed as a joint operation on quantifier se-541

quences and relations. See (Eijck & Unger, 2010, Chapter 10) for a full-fledged542

implementation and for further discussion.543

4 Example: Implementing Syllogistic Inference544

As an example of the process of implementing inference for natural language,545

let us view the language of the Aristotelian syllogism as a tiny fragment of546

natural language. Compare the chapter by Larry Moss on Natural Logic in547

this Handbook. The treatment in this Section is an improved version of the548

implementation in (Eijck & Unger, 2010, Chapter 5).549

The Aristotelian quantifiers are given in the following well-known square550

of opposition:551

All A are B No A are B

Some A are B Not all A are B
552

Aristotle interprets his quantifiers with existential import: All A are B553

and No A are B are taken to imply that there are A.554

What can we ask or state with the Aristotelian quantifiers? The following555

grammar gives the structure of queries and statements (with PN for plural556

nouns):557

Q ::= Are all PN PN?

| Are no PN PN?

| Are any PN PN?

| Are any PN not PN?

| What about PN?

558

S ::= All PN are PN.

| No PN are PN.

| Some PN are PN.

| Some PN are not PN.

The meanings of the Aristotelean quantifiers can be given in terms of set559

inclusion and set intersection, as follows:560

• ALL: Set inclusion561

• SOME: Non-empty set intersection562

• NOT ALL: Non-inclusion563

• NO: Empty intersection564

Set inclusion: A ⊆ B holds if and only if every element of A is an element565

of B. Non-empty set intersection: A ∩ B 6= ∅ if and only if there is some566

x ∈ A with x ∈ B. Non-empty set intersection can can expressed in terms of567

inclusion, negation and complementation, as follows: A∩B 6= ∅ if and only if568

A 6⊆ B.569

To get a sound and complete inference system for this, we use the following570

Key Fact: A finite set of syllogistic forms Σ is unsatisfiable if and only if571

there exists an existential form ψ such that ψ taken together with the universal572

forms from Σ is unsatisfiable.573

This restricted form of satisfiability can easily be tested with propositional574

logic. Suppose we talk about the properties of a single object x. Let proposition575

letter a express that object x has property A. Then a universal statement “All576

A are B” gets translated as a → b. An existential statement “Some A is B”577

gets translated as a ∧ b.578

For each property A we use a single proposition letter a. We have to check579

for each existential statement whether it is satisfiable when taken together580

with all universal statements. To test the satisfiability of a set of syllogistic581

statements with n existential statements we need n checks.582

Literals, Clauses, Clause Sets583

A literal is a propositional letter or its negation. A clause is a set of literals.584

A clause set is a set of clauses.585

Read a clause as a disjunction of its literals, and a clause set as a conjunc-586

tion of its clauses.587

Represent the propositional formula

(p→ q) ∧ (q → r)

as the following clause set:

{{¬p, q}, {¬q, r}}.

Here is an inference rule for clause sets: unit propagation588

Unit Propagation

If one member of a clause set is a singleton {l}, then:

• remove every other clause containing l from the clause set;
• remove l from every clause in which it occurs.

589

The result of applying this rule is a simplified equivalent clause set. For
example, unit propagation for {p} to

{{p}, {¬p, q}, {¬q, r}, {p, s}}

yields
{{p}, {q}, {¬q, r}}.

Applying unit propagation for {q} to this result yields:

{{p}, {q}, {r}}.

The Horn fragment of propositional logic consists of all clause sets where590

every clause has at most one positive literal. Satisfiability for syllogistic forms591

containing exactly one existental statement translates to the Horn fragment592

of propositional logic. HORNSAT is the problem of testing Horn clause sets593

for satisfiability. Here is an algorithm for HORNSAT:594

HORNSAT Algorithm

• If unit propagation yields a clause set in which units {l}, {l} occur, the
original clause set is unsatisfiable.

• Otherwise the units in the result determine a satisfying valuation.
Recipe: for all units {l} occurring in the final clause set, map their
proposition letter to the truth value that makes l true. Map all other
proposition letters to false.

595

Here is an implementation. The definition of literals:596

data Lit = Pos Name | Neg Name deriving Eq

instance Show Lit where

show (Pos x) = x

show (Neg x) = ’-’:x

neg :: Lit -> Lit

neg (Pos x) = Neg x

neg (Neg x) = Pos x

597

We can represent a clause as a list of literals:598

type Clause = [Lit]599

The names occurring in a list of clauses:600

names :: [Clause] -> [Name]

names = sort . nub . map nm . concat

where nm (Pos x) = x

nm (Neg x) = x

601

The implementation of the unit propagation algorithm: propagation of a602

single unit literal:603

unitProp :: Lit -> [Clause] -> [Clause]

unitProp x cs = concat (map (unitP x) cs)

unitP :: Lit -> Clause -> [Clause]

unitP x ys = if elem x ys then []

else

if elem (neg x) ys

then [delete (neg x) ys]

else [ys]

604

The property of being a unit clause:605

unit :: Clause -> Bool

unit [x] = True

unit _ = False

606

Propagation has the following type, where the Maybe expresses that the607

attempt to find a satisfying valuation may fail.608

propagate :: [Clause] -> Maybe ([Lit],[Clause])609

The implementation uses an auxiliary function prop with three arguments.610

The first argument gives the literals that are currently mapped to True, the611

second argument gives the literals that occur in unit clauses, the third argu-612

ment gives the non-unit clauses.613

propagate cls =

prop [] (concat (filter unit cls)) (filter (not.unit) cls)

where

prop :: [Lit] -> [Lit] -> [Clause]

-> Maybe ([Lit],[Clause])

prop xs [] clauses = Just (xs,clauses)

prop xs (y:ys) clauses =

if elem (neg y) xs

then Nothing

else prop (y:xs)(ys++newlits) clauses’ where

newclauses = unitProp y clauses

zs = filter unit newclauses

clauses’ = newclauses \\ zs

newlits = concat zs

614

Knowledge bases615

A knowledge base is a pair, with as first element the clauses that represent the616

universal statements, and as second element a lists of clause lists, consisting617

of one clause list per existential statement.618

type KB = ([Clause],[[Clause]])619

The intention is that the first element represents the universal statements,620

while the second element has one clause list per existential statement.621

The universe of a knowledge base is the list of all classes that are mentioned622

in it. We assume that classes are literals:623

type Class = Lit

universe :: KB -> [Class]

universe (xs,yss) =

map (\ x -> Pos x) zs ++ map (\ x -> Neg x) zs

where zs = names (xs ++ concat yss)

624

Statements and queries according to the grammar given above:625

data Statement =

All1 Class Class | No1 Class Class

| Some1 Class Class | SomeNot Class Class

| AreAll Class Class | AreNo Class Class

| AreAny Class Class | AnyNot Class Class

| What Class

deriving Eq

626

A statement display function is given in the appendix. Statement classifi-627

cation:628

isQuery :: Statement -> Bool

isQuery (AreAll _ _) = True

isQuery (AreNo _ _) = True

isQuery (AreAny _ _) = True

isQuery (AnyNot _ _) = True

isQuery (What _) = True

isQuery _ = False

629

Universal fact to statement. An implication p → q is represented as a630

clause {¬p, q}, and yields a universal statement “All p are q”. An implication631

p→ ¬q is represented as a clause {¬p,¬q}, and yields a statement “No p are632

q”.633

u2s :: Clause -> Statement

u2s [Neg x, Pos y] = All1 (Pos x) (Pos y)

u2s [Neg x, Neg y] = No1 (Pos x) (Pos y)

634

Existential fact to statement. A conjunction p∧q is represented as a clause635

set {{p}, {q}}, and yields an existential statement “Some p are q”. A conjunc-636

tion p ∧ ¬q is represented as a clause set {{p}, {¬q}}, and yields a statement637

“Some p are not q”.638

e2s :: [Clause] -> Statement

e2s [[Pos x],[Pos y]] = Some1 (Pos x) (Pos y)

e2s [[Pos x],[Neg y]] = SomeNot (Pos x) (Pos y)

639

Query negation:640

negat :: Statement -> Statement

negat (AreAll as bs) = AnyNot as bs

negat (AreNo as bs) = AreAny as bs

negat (AreAny as bs) = AreNo as bs

negat (AnyNot as bs) = AreAll as bs

641

The proper subset relation ⊂ is computed as the list of all pairs (x, y)642

such that adding clauses {x} and {¬y} — together these express that x ∩ y643

is non-empty — to the universal statements in the knowledge base yields644

inconsistency.645

subsetRel :: KB -> [(Class,Class)]

subsetRel kb =

[(x,y) | x <- classes, y <- classes,

propagate ([x]:[neg y]: fst kb) == Nothing ]

where classes = universe kb

646

If R ⊆ A2 and x ∈ A, then xR := {y | (x, y) ∈ R}. This is called a right647

section of a relation.648

rSection :: Eq a => a -> [(a,a)] -> [a]

rSection x r = [ y | (z,y) <- r, x == z ]
649

The supersets of a class are given by a right section of the subset relation,650

that is, the supersets of a class are all classes of which it is a subset.651

supersets :: Class -> KB -> [Class]

supersets cl kb = rSection cl (subsetRel kb)
652

The non-empty intersection relation is computed by combining each of the653

existential clause lists form the knowledge base with the universal clause list.654

intersectRel :: KB -> [(Class,Class)]

intersectRel kb@(xs,yys) =

nub [(x,y) | x <- classes, y <- classes, lits <- litsList,

elem x lits && elem y lits ]

where

classes = universe kb

litsList =

[ maybe [] fst (propagate (ys++xs)) | ys <- yys ]

655

The intersection sets of a class C are the classes that have a non-empty656

intersection with C:657

intersectionsets :: Class -> KB -> [Class]

intersectionsets cl kb = rSection cl (intersectRel kb)
658

In general, in KB query, there are three possibilities:659

(1) derive kb stmt is true. This means that the statement is derivable, hence660

true.661

(2) derive kb (neg stmt) is true. This means that the negation of stmt is662

derivable, hence true. So stmt is false.663

(3) neither derive kb stmt nor derive kb (neg stmt) is true. This means664

that the knowledge base has no information about stmt.665

The derivability relation is given by:666

derive :: KB -> Statement -> Bool

derive kb (AreAll as bs) = bs ‘elem‘ (supersets as kb)

derive kb (AreNo as bs) = (neg bs) ‘elem‘ (supersets as kb)

derive kb (AreAny as bs) = bs ‘elem‘ (intersectionsets as kb)

derive kb (AnyNot as bs) = (neg bs) ‘elem‘

(intersectionsets as kb)

667

To build a knowledge base we need a function for updating an existing668

knowledge base with a statement. If the update is successful, we want an669

updated knowledge base. If the update is not successful, we want to get an670

indication of failure. This explains the following type. The boolean in the671

output is a flag indicating change in the knowledge base.672

update :: Statement -> KB -> Maybe (KB,Bool)673

Update with an ‘All’ statement. The update function checks for possible674

inconsistencies. E.g., a request to add an A ⊆ B fact to the knowledge base675

leads to an inconsistency if A 6⊆ B is already derivable.676

update (All1 as bs) kb@(xs,yss)

| bs’ ‘elem‘ (intersectionsets as kb) = Nothing

| bs ‘elem‘ (supersets as kb) = Just (kb,False)

| otherwise = Just (([as’,bs]:xs,yss),True)

where

as’ = neg as

bs’ = neg bs

677

Update with other kinds of statements:678

update (No1 as bs) kb@(xs,yss)

| bs ‘elem‘ (intersectionsets as kb) = Nothing

| bs’ ‘elem‘ (supersets as kb) = Just (kb,False)

| otherwise = Just (([as’,bs’]:xs,yss),True)

where

as’ = neg as

bs’ = neg bs

679

update (Some1 as bs) kb@(xs,yss)

| bs’ ‘elem‘ (supersets as kb) = Nothing

| bs ‘elem‘ (intersectionsets as kb) = Just (kb,False)

| otherwise = Just ((xs,[[as],[bs]]:yss),True)

where

bs’ = neg bs

680

update (SomeNot as bs) kb@(xs,yss)

| bs ‘elem‘ (supersets as kb) = Nothing

| bs’ ‘elem‘ (intersectionsets as kb) = Just (kb,False)

| otherwise = Just ((xs,[[as],[bs’]]:yss),True)

where

bs’ = neg bs

681

The above implementation of an inference engine for syllogistic reasoning682

is a mini-case of computational semantics. What is the use of this? Cogni-683

tive research focusses on this kind of quantifier reasoning, so it is a pertinent684

question whether the engine can be used to meet cognitive realities? A possi-685

ble link with cognition would refine this calculus and the check whether the686

predictions for differences in processing speed for various tasks are realistic.687

There is also a link to the “natural logic for natural language” enterprise:688

the logical forms for syllogistic reasoning are very close to the surface forms689

of the sentences. The Chapter on Natural Logic in this Handbook gives more690

information. All in all, reasoning engines like this one are relevant for rational691

reconstructions of cognitive processing. The appendix gives the code for con-692

structing a knowledge base from a list of statements, and updating it. Here693

is a chat function that starts an interaction from a given knowledge base and694

writes the result of the interaction to a file:695

chat :: IO ()

chat = do

kb <- getKB "kb.txt"

writeKB "kb.bak" kb

putStrLn "Update or query the KB:"

str <- getLine

if str == "" then return ()

else do

handleCases kb str

chat

696

You are invited to try this out by loading the software for this chapter and697

running chat.698

5 Implementing Fragments of Natural Language699

Now what about the meanings of the sentences in a simple fragment of En-700

glish? Using what we know now about a logical form language and its inter-701

pretation in appropriate models, and assuming we have constants available for702

proper names, and predicate letters for the nouns and verbs of the fragment,703

we can easily translate the sentences generated by a simple example grammar704

into logical forms. Assume the following translation key:705

lexical item translation type of logical constant
girl Girl one-place predicate
boy Boy one-place predicate
toy Toy one-place predicate
laughed Laugh one-place predicate
cheered Cheer one-place predicate
loved Love two-place predicate
admired Admire two-place predicate
helped Help two-place predicate
defeated Defeat two-place predicate
gave Give three-place predicate
introduced Introduce three-place predicate
Alice a individual constant
Bob b individual constant
Carol c individual constant

706

Then the translation of Every boy loved a girl in the logical form language
above could become:

Q∀x(Boy x)(Q∃y(Girl y)(Love x y)).

To start the construction of meaning representations, we first represent707

a context free grammar for a natural language fragment in Haskell. A rule708

like S ::= NP VP defines syntax trees consisting of an S node immediately709

dominating an NP node and a VP node. This is rendered in Haskell as the710

following datatype definition:711

data S = S NP VP712

The S on the righthand side is a combinator indicating the name of the713

top of the tree. Here is a grammar for a tiny fragment:714

data S = S NP VP deriving Show

data NP = NP1 NAME | NP2 Det N | NP3 Det RN

deriving Show

data ADJ = Beautiful | Happy | Evil

deriving Show

data NAME = Alice | Bob | Carol

deriving Show

data N = Boy | Girl | Toy | N ADJ N

deriving Show

data RN = RN1 N That VP | RN2 N That NP TV

deriving Show

data That = That deriving Show

data VP = VP1 IV | VP2 TV NP | VP3 DV NP NP deriving Show

data IV = Cheered | Laughed deriving Show

data TV = Admired | Loved | Hated | Helped deriving Show

data DV = Gave | Introduced deriving Show

715

Look at this as a definition of syntactic structure trees. The structure for716

The boy that Alice helped admired every girl is given in Figure 1, with the717

Haskell version of the tree below it.718

Figure 1. Example structure tree

S

NP

Det

the

RN

N

boy

that NP

Alice

TV

helped

VP

TV

admired

NP

DET

every

CN

girl

S

(NP (Det the)

(RN (N boy) That (NP Alice) (TV helped))

(VP (TV admired) (NP (DET every) (N girl)))

For the purpose of this chapter we skip the definition of the parse function719

that maps the string The boy that Alice helped admired every girl to this720

structure (but see (Eijck & Unger, 2010, Chapter 9)).721

Now all we have to do is find appropriate translations for the categories in722

the grammar of the fragment. The first rule, S −→ NP VP, already presents723

us with a difficulty. In looking for NP translations and VP translations, should724

we represent NP as a function that takes a VP representation as argument,725

or vice versa?726

In any case, VP representations will have a functional type, for VPs de-727

note properties. A reasonable type for the function that represents a VP is728

Term -> Formula. If we feed it with a term, it will yield a logical form. Proper729

names now can get the type of terms. Take the example Alice laughed. The730

verb laughed gets represented as the function that maps the term x to the731

formula Atom "laugh" [x]. Therefore, we get an appropriate logical form for732

the sentence if x is a term for Alice.733

A difficulty with this approach is that phrases like no boy and every girl do734

not fit into this pattern. Following Montague, we can solve this by assuming735

that such phrases translate into functions that take VP representations as736

arguments. So the general pattern becomes: the NP representation is the737

function that takes the VP representation as its argument. This gives:738

lfS :: S -> Formula

lfS (S np vp) = (lfNP np) (lfVP vp)
739

Next, NP-representations are of type (Term -> Formula) -> Formula.740

lfNP :: NP -> (Term -> Formula) -> Formula

lfNP (NP1 Alice) = \ p -> p (Struct "Alice" [])

lfNP (NP1 Bob) = \ p -> p (Struct "Bob" [])

lfNP (NP1 Carol) = \ p -> p (Struct "Carol" [])

lfNP (NP2 det cn) = (lfDET det) (lfN cn)

lfNP (NP3 det rcn) = (lfDET det) (lfRN rcn)

741

Verb phrase representations are of type Term -> Formula.742

lfVP :: VP -> Term -> Formula

lfVP (VP1 Laughed) = \ t -> Atom "laugh" [t]

lfVP (VP1 Cheered) = \ t -> Atom "cheer" [t]

743

Representing a function that takes two arguments can be done either by744

means of a -> a -> b or by means of (a,a) -> b. A function of the first745

type is called curried, a function of the second type uncurried.746

We assume that representations of transitive verbs are uncurried, so they747

have type (Term,Term) -> Formula, where the first term slot is for the sub-748

ject, and the second term slot for the object. Accordingly, the representations749

of ditransitive verbs have type750

(Term,Term,Term) -> Formula751

where the first term slot is for the subject, the second one is for the indirect752

object, and the third one is for the direct object. The result should in both753

cases be a property for VP subjects. This gives us:754

lfVP (VP2 tv np) =

\ subj -> lfNP np (\ obj -> lfTV tv (subj,obj))

lfVP (VP3 dv np1 np2) =

\ subj -> lfNP np1 (\ iobj -> lfNP np2 (\ dobj ->

lfDV dv (subj,iobj,dobj)))

755

Representations for transitive verbs are:756

lfTV :: TV -> (Term,Term) -> Formula

lfTV Admired = \ (t1,t2) -> Atom "admire" [t1,t2]

lfTV Hated = \ (t1,t2) -> Atom "hate" [t1,t2]

lfTV Helped = \ (t1,t2) -> Atom "help" [t1,t2]

lfTV Loved = \ (t1,t2) -> Atom "love" [t1,t2]

757

Ditransitive verbs:758

lfDV :: DV -> (Term,Term,Term) -> Formula

lfDV Gave = \ (t1,t2,t3) -> Atom "give" [t1,t2,t3]

lfDV Introduced = \ (t1,t2,t3) ->

Atom "introduce" [t1,t2,t3]

759

Common nouns have the same type as VPs.760

lfN :: N -> Term -> Formula

lfN Girl = \ t -> Atom "girl" [t]

lfN Boy = \ t -> Atom "boy" [t]

761

The determiners we have already treated above, in Section 2. Complex762

common nouns have the same types as simple common nouns:763

lfRN :: RN -> Term -> Formula

lfRN (RN1 cn _ vp) = \ t -> Cnj [lfN cn t, lfVP vp t]

lfRN (RN2 cn _ np tv) = \ t -> Cnj [lfN cn t,

lfNP np (\ subj -> lfTV tv (subj,t))]

764

We end with some examples:765

lf1 = lfS (S (NP2 Some Boy)

(VP2 Loved (NP2 Some Girl)))

lf2 = lfS (S (NP3 No (RN2 Girl That (NP1 Bob) Loved))

(VP1 Laughed))

lf3 = lfS (S (NP3 Some (RN1 Girl That (VP2 Helped (NP1 Alice))))

(VP1 Cheered))

766

This gives:767

*IST> lf1768

Q Some x2 (Atom "boy" [x2])769

(Q Some x1 (Atom "girl" [x1]) (Atom "love" [x2,x1]))770

*IST> lf2771

Q No x1 (Cnj [Atom "girl" [x1],Atom "love" [Bob,x1]])772

(Atom "laugh" [x1])773

*IST> lf3774

Q Some x1 (Cnj [Atom "girl" [x1],Atom "help" [x1,Alice]])775

(Atom "cheer" [x1])776

What we have presented here is in fact an implementation of an exten-777

sional fragment of Montague grammar. The next Section indicates what has778

to change in an intensional fragment.779

6 Extension and Intension780

One of the trademarks of Montague grammar is the use of possible worlds to781

treat intensionality. Instead of giving a predicate a single interpretation in782

a model, possible world semantics gives intensional predicates different inter-783

pretations in different situations (or: in different “possible worlds”). A prince784

in one world may be a beggar in another, and the way in which intensional785

semantics accounts for this is by giving predicates like prince and beggar dif-786

ferent interpretations in different worlds.787

So we assume that apart from entities and truth values there is another788

basic type, for possible worlds. We introduce names or indices for possible789

worlds, as follows:790

data World = W Int deriving (Eq,Show)791

Now the type of individual concepts is the type of functions from worlds792

to entities, i.e., World -> Entity. An individual concept is a rigid designator793

if it picks the same entity in every possible world:794

rigid :: Entity -> World -> Entity

rigid x = \ _ -> x
795

A function from possible worlds to truth values is a proposition. Proposi-796

tions have type World -> Bool. In Mary desires to marry a prince the rigid797

designator that interprets the proper name “Mary” is related to a proposition,798

namely the proposition that is true in a world if and only if Mary marries799

someone who, in that world, is a prince. So an intensional verb like desire800

may have type (World -> Bool) -> (World -> Entity) -> Bool, where801

(World -> Bool) is the type of “marry a prince”, and (World -> Entity)802

is the type for the intensional function that interprets “Mary.”803

Models for intensional logic have a domain D of entities plus functions from804

predicate symbols to intensions of relations. Here is an example interpretion805

for the predicate symbol “princess:”806

princess :: World -> Pred Entity

princess = \ w [x] -> case w of

W 1 -> elem x [A,C,D,G]

W 2 -> elem x [A,M]

_ -> False

807

What this says is that in W1 x is a princess iff x is among A,C,D,G, in808

W2 x is a princess iff x is among A,M , and in no other world is x a princess.809

This interpretation for ”princess” will make “Mary is a princess” true in W2810

but in no other world.811

7 Implementing Communicative Action812

The simplest kind of communicative action probably is question answering of813

the kind that was demonstrated in the Syllogistics tool above, in Section 4.814

The interaction is between a system (the knowledge base) and a user. In the815

implementation we only keep track of changes in the system: the knowledge816

base gets updated every time the user makes statements that are consistent817

with the knowledge base but not derivable from it.818

Generalizing this, we can picture a group of communicating agents, each819

with their own knowledge, with acts of communication that change these820

knowledge bases. The basic logical tool for this is again intensional logic, more821

in particular the epistemic logic proposed by Hintikka in Hintikka (1962), and822

adapted in cognitive science (Gärdenfors (1988)), computer science (Fagin823

et al. (1995)) and economics (Aumann (1976); Battigalli & Bonanno (1999)).824

The general system for tracking how knowledge and belief of communicating825

agents evolve under various kinds of communication is called dynamic epis-826

temic logic or DEL. See van Benthem (2011) for a general perspective, and827

Ditmarsch et al. (2006) for a textbook account.828

To illustrate the basics, we will give an implementation of model checking829

for epistemic update logic with public announcements.830

The basic concept in the logic of knowledge is that of epistemic uncer-831

tainty. If I am uncertain about whether a coin that has just been tossed is832

showing head or tail, this can be pictured as two situations related by my833

uncertainty. Such uncertainty relations are equivalences: If I am uncertain be-834

tween situations s and t, and between situations t and r, this means I am also835

uncertain between s and r.836

Equivalence relations on a set of situations S can be implemented as parti-837

tions of S, where a partition is a family Xi of sets with the following properties838

(let I be the index set):839

• For each i ∈ I, Xi 6= ∅ and Xi ⊆ S.840

• For i 6= j, Xi ∩Xj = ∅.841

•
⋃

i∈I Xi = S.842

Here is a datatype for equivalence relations, viewed as partitions (lists of lists843

of items):844

type Erel a = [[a]]845

The block of an item x in a partition is the set of elements that are846

equivalent to x:847

bl :: Eq a => Erel a -> a -> [a]

bl r x = head (filter (elem x) r)
848

Page: 38 job: VanEijck macro: handbook.cls date/time: 8-Apr-2014/23:10



Implementing Semantic Theories 39

The restriction of a partition to a domain:849

restrict :: Eq a => [a] -> Erel a -> Erel a

restrict domain = nub . filter (/= [])

. map (filter (flip elem domain))

850

An infinite number of agents, with names a, b, c, d, e for the first five of851

them:852

data Agent = Ag Int deriving (Eq,Ord)

a,b,c,d,e :: Agent

a = Ag 0; b = Ag 1; c = Ag 2; d = Ag 3; e = Ag 4

instance Show Agent where

show (Ag 0) = "a"; show (Ag 1) = "b"; show (Ag 2) = "c";

show (Ag 3) = "d"; show (Ag 4) = "e";

show (Ag n) = ’a’: show n

853

A datatype for epistemic models:854

data EpistM state = Mo

[state]

[Agent]

[(Agent,Erel state)]

[state] deriving (Eq,Show)

855

An example epistemic model:856

example :: EpistM Int

example = Mo

[0..3]

[a,b,c]

[(a,[[0],[1],[2],[3]]),(b,[[0],[1],[2],[3]]),(c,[[0..3]])]

[1]

857

In this model there are three agents and four possible worlds. The first858

two agents a and b can distinguish all worlds, and the third agent c confuses859

all of them.860

Extracting an epistemic relation from a model:861

rel :: Agent -> EpistM a -> Erel a

rel ag (Mo _ _ rels _) = myLookup ag rels

myLookup :: Eq a => a -> [(a,b)] -> b

myLookup x table =

maybe (error "item not found") id (lookup x table)

862

This gives:863

*IST> rel a example864

[[0],[1],[2],[3]]865

*IST> rel c example866

[[0,1,2,3]]867

*IST> rel d example868

*** Exception: item not found869

A logical form language for epistemic statements; note that the type has870

a parameter for additional information.871

data Form a = Top

| Info a

| Ng (Form a)

| Conj [Form a]

| Disj [Form a]

| Kn Agent (Form a)

deriving (Eq,Ord,Show)

872

A useful abbreviation:873

impl :: Form a -> Form a -> Form a

impl form1 form2 = Disj [Ng form1, form2]
874

Semantic interpretation for this logical form language:875

isTrueAt :: Ord state =>

EpistM state -> state -> Form state -> Bool

isTrueAt m w Top = True

isTrueAt m w (Info x) = w == x

isTrueAt m w (Ng f) = not (isTrueAt m w f)

isTrueAt m w (Conj fs) = and (map (isTrueAt m w) fs)

isTrueAt m w (Disj fs) = or (map (isTrueAt m w) fs)

isTrueAt

m@(Mo worlds agents acc points) w (Kn ag f) = let

r = rel ag m

b = bl r w

in

and (map (flip (isTrueAt m) f) b)

876

This treats the Boolean connectives as usual, and interprets knowledge as877

truth in all worlds in the current accessible equivalence block of an agent.878

The effect of a public announcement φ on an epistemic model is that the879

set of worlds of that model gets limited to the worlds where φ is true, and the880

accessibility relations get restricted accordingly.881

upd_pa :: Ord state =>

EpistM state -> Form state -> EpistM state

upd_pa m@(Mo states agents rels actual) f =

(Mo states’ agents rels’ actual’)

where

states’ = [ s | s <- states, isTrueAt m s f ]

rels’ = [(ag,restrict states’ r) | (ag,r) <- rels ]

actual’ = [ s | s <- actual, s ‘elem‘ states’ ]

882

A series of public announcement updates:883

upds_pa :: Ord state =>

EpistM state -> [Form state] -> EpistM state

upds_pa m [] = m

upds_pa m (f:fs) = upds_pa (upd_pa m f) fs

884

We illustrate the working of the update mechanism on a famous epistemic885

puzzle. The following Sum and Product riddle was stated by the Dutch math-886

ematican Hans Freudenthal in a Dutch mathematics journal in 1969. There is887

also a version by John McCarthy (see http://www-formal.stanford.edu/888

jmc/puzzles.htm).889

A says to S and P: I have chosen two integers x, y such that 1 < x < y890

and x+ y ≤ 100. In a moment, I will inform S only of s = x+ y, and891

P only of p = xy. These announcements remain private. You are892

required to determine the pair (x, y). He acts as said. The following893

conversation now takes place:894

(1) P says: “I do not know the pair.”895

(2) S says: “I knew you didn’t.”896

(3) P says: “I now know it.”897

(4) S says: “I now also know it.”898

Determine the pair (x, y).899

This was solved by combinatorial means in a later issue of the journal. A900

model checking solution with DEMO Eijck (2007) (based on a DEMO program901

written by Ji Ruan) was presented in Ditmarsch et al. (2005). The present902

program is an optimized version of that solution.903

The list of candidate pairs:904

pairs :: [(Int,Int)]

pairs = [ (x,y) | x <- [2..100], y <- [2..100],

x < y, x+y <= 100 ]

905

The initial epistemic model is such that a (representing S) cannot dis-906

tinguish number pairs with the same sum, and b (representing P) cannot907

distinguish number pairs with the same product. Instead of using a valuation,908

we use number pairs as worlds.909

msnp :: EpistM (Int,Int)

msnp = (Mo pairs [a,b] acc pairs)

where

acc = [ (a, [ [ (x1,y1) | (x1,y1) <- pairs,

x1+y1 == x2+y2 ] |

(x2,y2) <- pairs ] ) ]

++

[ (b, [ [ (x1,y1) | (x1,y1) <- pairs,

x1*y1 == x2*y2 ] |

(x2,y2) <- pairs ] ) ]

910

The statement by b that he does not know the pair:911

statement_1 =

Conj [ Ng (Kn b (Info p)) | p <- pairs ]
912

To check this statement is expensive. A computationally cheaper equiva-913

lent statement is the following (see Ditmarsch et al. (2005)).914

statement_1e =

Conj [ Info p ‘impl‘ Ng (Kn b (Info p)) | p <- pairs ]
915

In Freudenthal’s story, the first public announcement is the statement916

where b confesses his ignorance, and the second public announcement is the917

statement by a about her knowledge about b’s state of knowledge before that918

confession. We can wrap the two together in a single statement to the effect919

that initially, a knows that b does not know the pair. This gives:920

k_a_statement_1e = Kn a statement_1e921

The second announcement proclaims the statement by b that now he922

knows:923

statement_2 =

Disj [ Kn b (Info p) | p <- pairs ]
924

Equivalently, but computationally more efficient:925

statement_2e =

Conj [ Info p ‘impl‘ Kn b (Info p) | p <- pairs ]
926

The final announcement concerns the statement by a that now she knows927

as well.928

statement_3 =

Disj [ Kn a (Info p) | p <- pairs ]
929

In the computationally optimized version:930

statement_3e =

Conj [ Info p ‘impl‘ Kn a (Info p) | p <- pairs ]
931

The solution:932

solution = upds_pa msnp

[k_a_statement_1e,statement_2e,statement_3e]
933

This is checked in a matter of minutes:934

*IST> solution935

Mo [(4,13)] [a,b] [(a,[[(4,13)]]),(b,[[(4,13)]])] [(4,13)]936

8 Resources937

Code for this Chapter938

The example code in this Chapter can be found at internet address https:939

//github.com/janvaneijck/ist. To run this software, you will need the940

Haskell system, which can be downloaded from www.haskell.org. This site941

also gives many interesting Haskell resources.942

Epistemic model checking943

More information on epistemic model checking can be found in the documen-944

tation of the epistemic model checker DEMO. See Eijck (2007).945

Link for Computational Semantics With Functional Programming946

The book Eijck & Unger (2010) has a website devoted to it, which can be947

found at www.computationalsemantics.eu.948

Further computational semantics links949

Special Interest Group in Computational Semantics: http://www.sigsem.950

org/wiki/. International Workshop on Computational Semantics: http:951

//iwcs.uvt.nl/. Wikipedia entry on computational semantics: http://en.952

wikipedia.org/wiki/Computational_semantics.953

9 Appendix954

A show function for identifiers:955

instance Show Id where

show (Id name 0) = name

show (Id name i) = name ++ show i

956

A show function for terms:957

instance Show Term where

show (Var id) = show id

show (Struct name []) = name

show (Struct name ts) = name ++ show ts

958

For the definition of fresh variables, we collect the list of indices that are959

used in the formulas in the scope of a quantifier, and select a fresh index, i.e.,960

an index that does not occur in the index list:961

fresh :: [Formula] -> Int

fresh fs = i+1 where i = maximum (0:indices fs)

indices :: [Formula] -> [Int]

indices [] = []

indices (Atom _ _:fs) = indices fs

indices (Eq _ _:fs) = indices fs

indices (Not f:fs) = indices (f:fs)

indices (Cnj fs1:fs2) = indices (fs1 ++ fs2)

indices (Dsj fs1:fs2) = indices (fs1 ++ fs2)

indices (Q _ (Id _ n) f1 f2:fs) = n : indices (f1:f2:fs)

962

A show function for the statements in our syllogistic inference fragment:963

instance Show Statement where

show (All1 as bs) =

"All " ++ show as ++ " are " ++ show bs ++ "."

show (No1 as bs) =

"No " ++ show as ++ " are " ++ show bs ++ "."

show (Some1 as bs) =

"Some " ++ show as ++ " are " ++ show bs ++ "."

show (SomeNot as bs) =

"Some " ++ show as ++ " are not " ++ show bs ++ "."

show (AreAll as bs) =

"Are all " ++ show as ++ show bs ++ "?"

show (AreNo as bs) =

"Are no " ++ show as ++ show bs ++ "?"

show (AreAny as bs) =

"Are any " ++ show as ++ show bs ++ "?"

show (AnyNot as bs) =

"Are any " ++ show as ++ " not " ++ show bs ++ "?"

show (What as) = "What about " ++ show as ++ "?"

964

Constructing a knowledge base from a list of statements:965

makeKB :: [Statement] -> Maybe KB

makeKB = makeKB’ ([],[])

where

makeKB’ kb [] = Just kb

makeKB’ kb (s:ss) = case update s kb of

Just (kb’,_) -> makeKB’ kb’ ss

Nothing -> Nothing

966

A preprocess function to prepare for parsing:967

preprocess :: String -> [String]

preprocess = words . (map toLower) .

(takeWhile (\ x -> isAlpha x || isSpace x))

968

A parse function, with a type indicating that the parsing may fail:969

parse :: String -> Maybe Statement

parse = parse’ . preprocess

where

parse’ ["all",as,"are",bs] =

Just (All1 (Pos as) (Pos bs))

parse’ ["no",as,"are",bs] =

Just (No1 (Pos as) (Pos bs))

parse’ ["some",as,"are",bs] =

Just (Some1 (Pos as) (Pos bs))

parse’ ["some",as,"are","not",bs] =

Just (SomeNot (Pos as) (Pos bs))

parse’ ["are","all",as,bs] =

Just (AreAll (Pos as) (Pos bs))

parse’ ["are","no",as,bs] =

Just (AreNo (Pos as) (Pos bs))

parse’ ["are","any",as,bs] =

Just (AreAny (Pos as) (Pos bs))

parse’ ["are","any",as,"not",bs] =

Just (AnyNot (Pos as) (Pos bs))

parse’ ["what", "about", as] = Just (What (Pos as))

parse’ ["how", "about", as] = Just (What (Pos as))

parse’ _ = Nothing

970

Processing a piece of text, given as a string with newline characters.971

process :: String -> KB

process txt =

maybe ([],[]) id (mapM parse (lines txt) >>= makeKB)

972

An example text, consisting of lines separated by newline characters:973

mytxt = "all bears are mammals\n"

++ "no owls are mammals\n"

++ "some bears are stupids\n"

++ "all men are humans\n"

++ "no men are women\n"

++ "all women are humans\n"

++ "all humans are mammals\n"

++ "some men are stupids\n"

++ "some men are not stupids"

974

Reading a knowledge base from disk:975

getKB :: FilePath -> IO KB

getKB p = do

txt <- readFile p

return (process txt)

976

Writing a knowledge base to disk, in the form of a list of statements.977

writeKB :: FilePath -> KB -> IO ()

writeKB p (xs,yss) = writeFile p (unlines (univ ++ exist))

where

univ = map (show.u2s) xs

exist = map (show.e2s) yss

978

Telling about a class, based on the info in a knowledge base.979

tellAbout :: KB -> Class -> [Statement]

tellAbout kb as =

[All1 as (Pos bs) | (Pos bs) <- supersets as kb,

as /= (Pos bs) ]

++

[No1 as (Pos bs) | (Neg bs) <- supersets as kb,

as /= (Neg bs) ]

++

[Some1 as (Pos bs) | (Pos bs) <- intersectionsets as kb,

as /= (Pos bs),

notElem (as,Pos bs) (subsetRel kb) ]

++

[SomeNot as (Pos bs) | (Neg bs) <- intersectionsets as kb,

notElem (as, Neg bs) (subsetRel kb) ]

980

Depending on the input, the various cases are handled by the following981

function:982

handleCases :: KB -> String -> IO ()

handleCases kb str =

case parse str of

Nothing -> putStrLn "Wrong input.\n"

Just (What as) -> let

info = (tellAbout kb as, tellAbout kb (neg as)) in

case info of

([],[]) -> putStrLn "No info.\n"

([],negi) -> putStrLn (unlines (map show negi))

(posi,negi) -> putStrLn (unlines (map show posi))

Just stmt ->

if isQuery stmt then

if derive kb stmt then putStrLn "Yes.\n"

else if derive kb (negat stmt)

then putStrLn "No.\n"

else putStrLn "I don’t know.\n"

else case update stmt kb of

Just (kb’,True) -> do

writeKB "kb.txt" kb’

putStrLn "OK.\n"

Just (_,False) -> putStrLn

"I knew that already.\n"

Nothing -> putStrLn

"Inconsistent with my info.\n"

983

